
Upgrading to 2.0
    

A 2.0 release was published with a minor breaking change. We decided to vendor NimbleOptions (copy their code into our codebase) so that we could make some necessary modifications to it. What this means for users is primarily that:
	we no longer depend on NimbleOptions
	if you are matching on NimbleOptions.ValidationError you will need to update your code to match on Spark.Options.ValidationError



  

    
Writing Extensions
    

Writing extensions generally involves three main components.

  
    
  
  The DSL declaration


The DSL is declared as a series of Spark.Dsl.Section, which can contain Spark.Dsl.Entity and further Spark.Dsl.Section structs. See Spark.Dsl.Section and Spark.Dsl.Entity for more information.

  
    
  
  Transformers


Extension writing gets a bit more complicated when you get into the world of transformers, but this is also where a lot of the power is. Each transformer can declare other transformers it must go before or after, and then is given the opportunity to modify the entirety of the DSL it is extending up to that point. This allows extensions to make rich modifications to the structure in question. See Spark.Dsl.Transformer for more information

  
    
  
  Introspection


Use functions in Spark.Dsl.Extension to retrieve the stored values from the DSL and expose them in a module. The convention is to place functions for something like MyApp.MyExtension in MyApp.MyExtension.Info. Using introspection functions like this allows for a richer introspection API (i.e not just getting and retrieving raw values), and it also allows us to add type specs and documentation, which is helpful when working generically. I.e module_as_variable.table() can't be known by dialyzer, whereas Extension.table(module) can be.


  

    
Splitting Up Large DSLs
    

When building large DSLs, we face similar problems as things like large configuration files. It can be hard to find what we're looking for, and we can end up scrolling through a lot of DSL code to find what we're interested in. We generally suggest avoiding splitting up your DSLs by default, but it is important to know how to do so when the need arises.

  
    
  
  Fragments


Spark offers a tool called Spark.Dsl.Fragment, which allows you to compose a single DSL from multiple smaller DSL modules. There are a few important properties and caveats to understand:
	Fragments are not designed for sharing code between instances of a spark DSL. They are not dynamic. For creating behavior that extends across multiple instances of a DSL, you should write an extension.

	A DSL has all extensions that any of its fragments has.

	Fragments must express what they are a fragment of.



  
    
  
  Example


defmodule MyApp.Accounts.User.Fragments.DataLayer do
  use Spark.Dsl.Fragment,
    of: Ash.Resource,
    data_layer: AshPostgres.DataLayer

  postgres do
    table "users"
    repo MyApp.Repo
    ...
  end
end

defmodule MyApp.Accounts.User do
  use Ash.Resource,
    fragments: [MyApp.Accounts.User.Fragments.DataLayer]

  ...
end


  

    
Spark
    

Spark helps you build powerful and well documented DSLs that come with useful tooling out of the box. DSLs are declared using simple structs, and every DSL has the ability to be extended by the end user. Spark powers all of the DSLs in Ash Framework.
What you get for your DSL when you implement it with Spark:
	Extensibility. Anyone can write extensions for your DSL.
	Autocomplete and in-line documentation: An elixir_sense plugin that "Just Works" for any DSL implemented with Spark.
	Tools to generate documentation for your DSL automatically.
	A mix task to add every part of the DSL to the locals_without_parens of your library automatically.

This library still needs a lot of documentation. It is very stable as it is used for all of the Ash packages,
but PRs are very welcome for more documentation and examples.

  
    
  
  Dependency


{:spark, "~> 2.2.42"}

  
    
  
  Your First DSL


Lets define a simple DSL that can be used to define data validators. We will call it MyLibrary.Validator.
You will be able to use it like so:
defmodule MyApp.PersonValidator do
  use MyLibrary.Validator

  fields do
    required [:name]
    field :name, :string

    field :email, :string do
      check &String.contains?(&1, "@")
      process &String.trim/1
    end

    # This syntax is also supported
    # field :email, :string, check: &String.contains?(&1, "@"), process: &String.trim/1
  end
end

MyApp.PersonValidator.validate(%{name: "Zach", email: " foo@example.com "})
{:ok, %{name: "Zach", email: "foo@example.com"}}

MyApp.PersonValidator.validate(%{name: "Zach", email: " blank "})
:error
There are many ways you can enhance this data validator, but we are using this only as a simple realistic example.

  
    
  
  Defining the DSL extension


We define our DSL as a Spark.Dsl.Extension. All DSLs are defined in extensions, but you can define DSLs that use
certain extensions by default, which is what we will do here.
Here we are building up one big nested data structure of a %Spark.Dsl.Section{}, which spark
uses under the hood to allow users to write in the DSL syntax described above.
defmodule MyLibrary.Validator.Dsl do
  defmodule Field do
    defstruct [:name, :type, :transform, :check]
  end

  @field %Spark.Dsl.Entity{
    name: :field,
    args: [:name, :type],
    target: Field,
    describe: "A field that is accepted by the validator",
    # you can include nested entities here, but
    # note that you provide a keyword list like below
    # we need to know which struct key to place the nested entities in
    # entities: [
    #   key: [...]
    # ],
    schema: [
      name: [
        type: :atom,
        required: true,
        doc: "The name of the field"
      ],
      type: [
        type: {:one_of, [:integer, :string]},
        required: true,
        doc: "The type of the field"
      ],
      check: [
        type: {:fun, 1},
        doc: "A function that can be used to check if the value is valid after type validation."
      ],
      transform: [
        type: {:fun, 1},
        doc: "A function that will be used to transform the value after successful validation"
      ]
    ]
  }

  @fields %Spark.Dsl.Section{
    name: :fields,
    schema: [
      required: [
        type: {:list, :atom},
        doc: "The fields that must be provided for validation to succeed"
      ]
    ],
    entities: [
      @field
    ],
    describe: "Configure the fields that are supported and required"
  }

  use Spark.Dsl.Extension, sections: [@fields]
end

  
    
  
  Defining the DSL module


Now lets define our actual DSL module. This is what people will use when they want to define a module that uses our DSL.
defmodule MyLibrary.Validator do
  use Spark.Dsl,
    default_extensions: [
      extensions: [MyLibrary.Validator.Dsl]
    ]
end
Now, we can define something using this DSL!
defmodule MyApp.PersonValidator do
  use MyLibrary.Validator

  fields do
    required [:name]
    field :name, :string

    field :email, :string do
      check &String.contains?(&1, "@")
      transform &String.trim/1
    end

    # This syntax is also supported
    # field :email, :string, check: &String.contains?(&1, "@"), transform: &String.trim/1
  end
end

  
    
  
  Getting information out of our DSL


For this, we can use various functions in Spark.Dsl.Extension. For example:
iex(1)> Spark.Dsl.Extension.get_entities(MyApp.PersonValidator, :fields)
iex(2)> Spark.Dsl.Extension.get_entities(MyApp.PersonValidator, :fields)
[
  %MyLibrary.Validator.Dsl.Field{
    name: :name,
    type: :string,
    transform: nil,
    check: nil
  },
  %MyLibrary.Validator.Dsl.Field{
    name: :email,
    type: :string,
    transform: &String.trim/1,
    # This is an example of some under the hood magic that spark does
    # to allow you to define a function inside your DSL. This sort of thing
    # is quite difficult to do with hand-rolled DSLs.
    check: &MyApp.PersonValidator.check_0_generated_18E6D5D8C34DFA0EDA8E926DAAEE7E52/1
  }
]

  
    
  
  Getting a nice interface to your DSL


Spark provides a nice tool called InfoGenerator which defines functions for you automatically
corresponding to your DSL. Lets give it a whirl. The general pattern is to define a module called
YourDsl.Info for this.
defmodule MyLibrary.Validator.Info do
  use Spark.InfoGenerator, extension: MyLibrary.Validator.Dsl, sections: [:fields]
end
Which can be used like so:
iex(1)> MyLibrary.Validator.Info.fields(MyApp.PersonValidator)
[
  %MyLibrary.Validator.Dsl.Field{
    name: :name,
    type: :string,
    transform: nil,
    check: nil
  },
  %MyLibrary.Validator.Dsl.Field{
    name: :email,
    type: :string,
    transform: &String.trim/1,
    check: &MyApp.PersonValidator.check_0_generated_18E6D5D8C34DFA0EDA8E926DAAEE7E52/1
  }
]
# Returns `:error` for fields not specified
iex(2)> MyLibrary.Validator.Info.fields_required(MyApp.PersonValidator)
{:ok, [:name]}
# The `!` version can be used for fields you know will always be set
iex(3)> MyLibrary.Validator.Info.fields_required!(MyApp.PersonValidator)
[:name]

  
    
  
  Transformers


Transformers are an extremely powerful concept in Spark. They allow for arbitrary transformations
of the data structure backing our Dsl at compile time. This actually allows us to avoid a lot of
magic that you see in macro-based DSLs. We can write simple, regular Elixir code! For instance,
lets say we want to add an :id field to all validators.
defmodule MyLibrary.Validator.Transformers.AddId do
  use Spark.Dsl.Transformer

  # dsl_state here is a map of the underlying DSL data
  def transform(dsl_state) do
    {:ok,
      Spark.Dsl.Transformer.add_entity(dsl_state, [:fields], %MyLibrary.Validator.Dsl.Field{
        name: :id,
        type: :string
      })
    }
  end
end
Now, we can add this transformer to our DSL extension. Modify your extension like so
use Spark.Dsl.Extension, sections: [@fields], transformers: [
  MyLibrary.Validator.Transformers.AddId
]
Now our DSLs will all have an :id field automatically. If we recompile, we can see this in action:
iex(1)> MyLibrary.Validator.Info.fields(MyApp.PersonValidator)
[
  %MyLibrary.Validator.Dsl.Field{
    name: :id,
    type: :string,
    transform: nil,
    check: nil
  },
  %MyLibrary.Validator.Dsl.Field{
    name: :name,
    type: :string,
    transform: nil,
    check: nil
  },
  %MyLibrary.Validator.Dsl.Field{
    name: :email,
    type: :string,
    transform: &String.trim/1,
    check: &MyApp.PersonValidator.check_0_generated_18E6D5D8C34DFA0EDA8E926DAAEE7E52/1
  }
]

  
    
  
  Verifiers


Verifiers are similar to transformers, except that they cannot modify the structure. They can
only return :ok or {:error, error}. This is important because when verifiers are running
you know that you are looking at the final structure of the DSL. Prefer to write verifiers
over transformers if you are only doing some kind of validation.
Lets make a verifier that says that all fields in required must also be in fields.
defmodule MyLibrary.Validator.Verifiers.VerifyRequired do
  use Spark.Dsl.Verifier

  # dsl_state here is a map of the underlying DSL data
  def verify(dsl_state) do
    # we can use our info module here, even though we are passing in a
    # map of data and not a module! Very handy.

    required = MyLibrary.Validator.Info.fields_required!(dsl_state)
    fields = Enum.map(MyLibrary.Validator.Info.fields(dsl_state), &(&1.name))

    if Enum.all?(required, &Enum.member?(fields, &1)) do
      :ok
    else
      {:error,
       Spark.Error.DslError.exception(
         message: "All required fields must be specified in fields",
         path: [:fields, :required],
         # this is how you get the original module out.
         # only do this for display purposes.
         # the module is not yet compiled (we're compiling it right now!), so if you
         # try to call functions on it, you will deadlock the compiler
         # and get an error
         module: Spark.Dsl.Verifier.get_persisted(dsl_state, :module)
       )}
    end
  end
end
Now we can include this in our DSL extension as well!
use Spark.Dsl.Extension,
  sections: [@fields],
  transformers: [
    MyLibrary.Validator.Transformers.AddId
  ],
  verifiers: [
    MyLibrary.Validator.Verifiers.VerifyRequired
  ]
Now if someone tries to define an invalid validator (yo dawg I heard you like validation),
they will get a nice error message:
defmodule MyApp.BadValidator do
  use MyLibrary.Validator

  fields do
    required [:name, :email]
    field :name, :string
  end
end
produces
** (Spark.Error.DslError) [MyApp.BadValidator]
fields -> required:
  All required fields must be specified in fields
    (elixir 1.17.2) lib/process.ex:864: Process.info/2
    (spark 2.2.35) lib/spark/error/dsl_error.ex:30: Spark.Error.DslError.exception/1
    iex:54: MyLibrary.Validator.Verifiers.VerifyRequired.verify/1
    iex:40: anonymous fn/1 in MyApp.BadValidator.__verify_spark_dsl__/1
    (elixir 1.17.2) lib/enum.ex:4353: Enum.flat_map_list/2
    (elixir 1.17.2) lib/enum.ex:4354: Enum.flat_map_list/2
    iex:40: MyApp.BadValidator.__verify_spark_dsl__/1
    (elixir 1.17.2) lib/enum.ex:987: Enum."-each/2-lists^foreach/1-0-"/2
In the future we will add support for including location information in these errors,
by allowing you to look up where a given entity/option was defined in the source code,
so the user gets nice squiggly lines in their editor.

  
    
  
  Generating code into the module


We want each validator to have a validate function, so we need to generate some code.
The best way to do that is with another transformer.
defmodule MyLibrary.Validator.Transformers.GenerateValidate do
  use Spark.Dsl.Transformer

  def transform(dsl_state) do

    validate = quote do
      def validate(data) do
        # Our generated code can be very simple
        # because we can get all the info we need from the module
        # in our regular ELixir code.
        MyLibrary.Validator.validate(__MODULE__, data)
      end
    end

    {:ok, Spark.Dsl.Transformer.eval(dsl_state, [], validate)}
  end
end
Now we can include this in our DSL, and then define validate/2,
and we're done!
use Spark.Dsl.Extension,
  sections: [@fields],
  transformers: [
    MyLibrary.Validator.Transformers.AddId,
    MyLibrary.Validator.Transformers.GenerateValidate
  ],
  verifiers: [
    MyLibrary.Validator.Verifiers.VerifyRequired
  ]
Now we can define our validation implementation:
defmodule MyLibrary.Validator do
  use Spark.Dsl,
    default_extensions: [
      extensions: [MyLibrary.Validator.Dsl]
    ]

  def validate(module, data) do
    fields = MyLibrary.Validator.Info.fields(module)
    required = MyLibrary.Validator.Info.fields_required!(module)

    case Enum.reject(required, &Map.has_key?(data, &1)) do
      [] ->
        validate_fields(fields, data)
      missing_required_fields ->
        {:error, :missing_required_fields, missing_required_fields}
    end
  end

  defp validate_fields(fields, data) do
    Enum.reduce_while(fields, {:ok, %{}}, fn field, {:ok, acc} ->
      case Map.fetch(data, field.name) do
        {:ok, value} ->
          case validate_value(field, value) do
            {:ok, value} ->
              {:cont, {:ok, Map.put(acc, field.name, value)}}
            :error ->
              {:halt, {:error, :invalid, field.name}}
          end

        :error ->
          {:cont, {:ok, acc}}
    end)
  end

  defp validate_value(field, value) do
    with true <- type_check(field, value),
         true <- check(field, value) do
      {:ok, transform(field, value)}
    else
      _ ->
        :error
    end
  end

  defp type_check(%{type: :string}, value) when is_binary(value) do
    true
  end

  defp type_check(%{type: :integer}, value) when is_integer(value) do
    true
  end

  defp type_check(_, _), do: false

  defp check(%{check: check}, value) when is_function(check, 1) do
    check.(value)
  end

  defp check(_, _), do: true

  defp transform(%{transform: transform}, value) when is_function(transform, 1) do
    transform.(value)
  end

  defp transform(_, value), do: value
end

  
    
  
  Conclusion


There is a lot more to spark, but this example shows you the kinds of things that you can build,
as well as the "programming model" around spark. Specify a DSL, define instances of that DSL,
inspect those instances to do things like generate code or write functions like validate/2 above!


  

    
Spark.Dsl behaviour
    



      
The primary entry point for defining a DSL.
To define a DSL, add use Spark.Dsl, ...options. The options supported with use Spark.Dsl are:
	:single_extension_kinds (list of atom/0) - The extension kinds that are allowed to have a single value. For example: [:data_layer] The default value is [].

	:many_extension_kinds (list of atom/0) - The extension kinds that can have multiple values. e.g [notifiers: [Notifier1, Notifier2]] The default value is [].

	:untyped_extensions? (boolean/0) - Whether or not to support an extensions key which contains untyped extensions The default value is true.

	:extension_kind_types (keyword/0) - A keyword list of extension kinds and their types, e.g [authorizers: {:list, {:behaviour, Ash.Authorizer}}] The default value is [].

	:extension_kind_docs (keyword/0) - A keyword list of extension kinds and a short documentation snippet to be used when autocompleting that option The default value is [].

	:default_extensions (keyword/0) - The extensions that are included by default. e.g [data_layer: Default, notifiers: [Notifier1]]
Default values for single extension kinds are overwritten if specified by the implementor, while many extension
kinds are appended to if specified by the implementor. The default value is [].

	:opt_schema (keyword/0) - A schema for additional options to accept when calling use YourSpark The default value is [].

	:opts_to_document - A list of atom/0 or :all. Spark automatically detects options and documents them in @moduledoc.
You can instruct Spark to use only a subset of options, e.g. opts_to_document: [:fragments]. The default value is :all.


See the callbacks defined in this module to augment the behavior/compilation of the module getting a Dsl.

  
    
  
  Schemas/Data Types


For more information, see Spark.Options.

    Spark.Dsl.Builder - spark v2.2.42
    
    

    


  
  

    
Spark.Dsl.Builder 
    



      
Utilities for building DSL objects programatically, generally used in transformers.

Spark.Dsl.Entity 
    



      
Declares a DSL entity.
A dsl entity represents a dsl constructor who's resulting value is a struct.
This lets the user create complex objects with arbitrary(mostly) validation rules.
The lifecycle of creating entities is complex, happening as Elixir is compiling
the modules in question. Some of the patterns around validating/transforming entities
have not yet solidified. If you aren't careful and don't follow the guidelines listed
here, you can have subtle and strange bugs during compilation. Anything not isolated to
simple value validations should be done in transformers. See Spark.Dsl.Transformer.
An entity has a target indicating which struct will ultimately be built. An entity
also has a schema. This schema is used for documentation, and the options are validated
against it before continuing on with the DSL.
To create positional arguments to the builder, use args. The values provided to
args need to be in the provided schema as well. They will be positional arguments
in the same order that they are provided in the args key.
auto_set_fields will set the provided values into the produced struct (they do not need
to be included in the schema).
transform is a function that takes a created struct and can alter it. This happens immediately
after handling the DSL options, and can be useful for setting field values on a struct based on
other values in that struct. If you need things that aren't contained in that struct, use an
Spark.Dsl.Transformer. This function returns {:ok, new_entity} or {:error, error}, so this can
also be used to validate the entity.
entities allows you to specify a keyword list of nested entities. Nested entities are stored
on the struct in the corresponding key, and are used in the same way entities are otherwise.
singleton_entity_keys specifies a set of entity keys (specified above) that should only have a
single value. This will be validated and unwrapped into nil | single_value on success.
identifier expresses that a given entity is unique by that field, validated by the DSL.

  
    
  
  Example


@my_entity %Spark.Dsl.Entity{
  name: :my_entity,
  target: MyStruct,
  schema: [my_field: [type: :atom, required: false]]
}
Once compiled by Spark, entities can be invoked with a keyword list:
my_entity my_field: :value
Or with a do block:
my_entity do
  my_field :value
end
For a full example, see Spark.Dsl.Extension.

Spark.Dsl.Extension behaviour
    



      
An extension to the Spark DSL.
This allows configuring custom DSL components, whose configurations
can then be read back. This guide is still a work in progress, but should
serve as a decent example of what is possible. Open issues on Github if you
have any issues/something is unclear.
The example at the bottom shows how you might build a (not very contextually
relevant) DSL extension that would be used like so:
defmodule MyApp.Vehicle do
  use Spark.Dsl
end

defmodule MyApp.MyResource do
  use MyApp.Vehicle,
    extensions: [MyApp.CarExtension]

  cars do
    car :ford, :focus, trim: :sedan
    car :toyota, :corolla
  end
end
The extension:
defmodule MyApp.CarExtension do
  @car_schema [
    make: [
      type: :atom,
      required: true,
      doc: "The make of the car"
    ],
    model: [
      type: :atom,
      required: true,
      doc: "The model of the car"
    ],
    type: [
      type: :atom,
      required: true,
      doc: "The type of the car",
      default: :sedan
    ]
  ]

  @car %Spark.Dsl.Entity{
    name: :car,
    describe: "Adds a car",
    examples: [
      "car :ford, :focus"
    ],
    target: MyApp.Car,
    args: [:make, :model],
    schema: @car_schema
  }

  @cars %Spark.Dsl.Section{
    name: :cars, # The DSL constructor will be `cars`
    describe: """
    Configure what cars are available.

    More, deeper explanation. Always have a short one liner explanation,
    an empty line, and then a longer explanation.
    """,
    entities: [
      @car # See `Spark.Dsl.Entity` docs
    ],
    schema: [
      default_manufacturer: [
        type: :atom,
        doc: "The default manufacturer"
      ]
    ]
  }

  use Spark.Dsl.Extension, sections: [@cars]
end
Often, we will need to do complex validation/validate based on the configuration
of other resources. Due to the nature of building compile time DSLs, there are
many restrictions around that process. To support these complex use cases, extensions
can include transformers which can validate/transform the DSL state after all basic
sections/entities have been created. See Spark.Dsl.Transformer for more information.
Transformers are provided as an option to use, like so:
use Spark.Dsl.Extension, sections: [@cars], transformers: [
  MyApp.Transformers.ValidateNoOverlappingMakesAndModels
]
By default, the generated modules will have names like __MODULE__.SectionName.EntityName, and that could
potentially conflict with modules you are defining, so you can specify the module_prefix option, which would allow
you to prefix the modules with something like __MODULE__.Dsl, so that the module path generated might be something like
__MODULE__.Dsl.SectionName.EntityName, and you could then have the entity struct be __MODULE__.SectionName.EntityName
without conflicts.
To expose the configuration of your DSL, define functions that use the
helpers like get_entities/2 and get_opt/3. For example:
defmodule MyApp.Cars do
  def cars(resource) do
    Spark.Dsl.Extension.get_entities(resource, [:cars])
  end
end

MyApp.Cars.cars(MyResource)
# [%MyApp.Car{...}, %MyApp.Car{...}]
See the documentation for Spark.Dsl.Section and Spark.Dsl.Entity for more information

Spark.Dsl.Fragment 
    



      
Allows splitting up a DSL into multiple modules, potentially organizing large DSLs
Use the of option to expression what your fragment is a fragment of. You can add
extensions as you would normally to that resource, and they will be added to the
parent resource.
defmodule MyApp.Resource.Graphql do
  use Spark.Dsl.Fragment, of: Ash.Resource, extensions: AshGraphql.Resource

  graphql do
    ...
  end
end
Then add the fragment to the parent resource.
defmodule MyApp.Resource do
  use Ash.Resource, fragments: [MyApp.Resource.Graphql], ...
end

Spark.Dsl.Patch.AddEntity 
    



      
Supply this when defining an extension to add entity builders to another extension's section.
For example
@entity %Spark.Dsl.Entity{
  ...
}

@dsl_patch %Spark.Dsl.Patch.AddEntity{section_path: [:foo, :bar], entity: @entity}

use Spark.Dsl.Extension, dsl_patches: [@dsl_patch]

Spark.Dsl.Section 
    



      
Declares a DSL section.
A dsl section allows you to organize related configurations. All extensions
configure sections, they cannot add DSL builders to the top level. This
keeps things organized, and concerns separated.
A section may have nested sections, which will be configured the same as other sections.
Getting the options/entities of a section is done by providing a path, so you would
use the nested path to retrieve that configuration. See Spark.Dsl.Extension.get_entities/2
and Spark.Dsl.Extension.get_opt/4.
A section may have entities, which are constructors that produce instances of structs.
For more on entities, see Spark.Dsl.Entity.
A section may also have a schema, which you can learn more about in Spark.Options. Spark will produce
builders for those options, so that they may be configured. They are retrieved with
Spark.Dsl.Extension.get_opt/4.
To create a section that is available at the top level (i.e not  nested inside of its own name), use
top_level?: true. Remember, however, that this has no effect on sections nested inside of other sections.
For a full example, see Spark.Dsl.Extension.

Spark.Dsl.Transformer behaviour
    



      
A transformer manipulates and/or validates the entire DSL state of a resource.
It's transform/1 takes a map, which is just the values/configurations at each point
of the DSL. Don't manipulate it directly, if possible, instead use functions like
get_entities/3 and replace_entity/4 to manipulate it.
Use the after?/1 and before?/1 callbacks to ensure that your transformer
runs either before or after some other transformer.
Return true in after_compile/0 to have the transformer run in an after_compile hook,
but keep in mind that no modifications to the dsl structure will be retained, so there is no
real point in modifying the dsl that you return.

Spark.Dsl.Verifier behaviour
    



      
A verifier gets the dsl state and can return :ok or :error.
In a verifier, you can reference and depend on other modules without causing compile time dependencies.

Spark.Dsl.Verifiers.VerifyEntityUniqueness 
    



      
Verifies that each entity that has an identifier is unique at each path.

Spark.Options 
    



      
Provides a standard API to handle keyword-list-based options.
This module began its life as a vendored form of NimbleOptions,
meaning that we copied it from NimbleOptions into Spark.
We had various features to add to it, and the spirit of nimble
options is to be as lightweight as possible. With that in mind,
we were advised to vendor it. We would like to thank the authors
of NimbleOptions for their excellent work, and their blessing
to transplant their work into Spark.
Spark.Options allows developers to create schemas using a
pre-defined set of options and types. The main benefits are:
	A single unified way to define simple static options
	Config validation against schemas
	Automatic doc generation
	More types over what is provided by NimbleOptions
	Compile time validators that are highly optimized and produce structs. See Spark.Options.Validator.
	Shared logic between Spark DSLs and options lists.


  
    
  
  Schema Options


These are the options supported in a schema. They are what
defines the validation for the items in the given schema.
	:type - The type of the option item. The default value is :any.

	:required (boolean/0) - Defines if the option item is required. The default value is false.

	:default (term/0) - The default value for the option item if that option is not specified. This value
is validated according to the given :type. This means that you cannot
have, for example, type: :integer and use default: "a string".

	:keys (keyword/0) - Available for types :keyword_list, :non_empty_keyword_list, and :map,
it defines which set of keys are accepted for the option item. The value of the
:keys option is a schema itself. For example: keys: [foo: [type: :atom]].
Use :* as the key to allow multiple arbitrary keys and specify their schema:
keys: [*: [type: :integer]].

	:deprecated (String.t/0) - Defines a message to indicate that the option item is deprecated. The message will be displayed as a warning when passing the item.

	:private? (boolean/0) - Defines an option as private, used with Spark.Options.Validator The default value is false.

	:hide (one or a list of atom/0) - A list of keys that should be hidden when generating documentation

	:as (atom/0) - A name to remap the option to when used in DSLs. Not supported in regular option parsing

	:snippet (String.t/0) - A snippet to use when autocompleting DSLs. Not supported in regular option parsing

	:links (term/0) - A keyword list of links to include in DSL documentation for the option item.

	:doc (String.t/0 or false) - The documentation for the option item.

	:subsection (String.t/0) - The title of separate subsection of the options' documentation

	:type_doc (String.t/0 or false) - The type doc to use in the documentation for the option item. If false,
no type documentation is added to the item. If it's a string, it can be
anything. For example, you can use "a list of PIDs", or you can use
a typespec reference that ExDoc can link to the type definition, such as
"`t:binary/0`". You can use Markdown in this documentation. If the
:type_doc option is not present, Spark.Options tries to produce a type
documentation automatically if it can do it unambiguously. For example,
if type: :integer, Spark.Options will use integer/0 as the
auto-generated type doc.

	:type_spec (Macro.t/0) - The quoted spec to use in the typespec for the option item. You should use this
when the auto-generated spec is not specific enough. For example, if you are performing
custom validation on an option (with the {:custom, ...} type), then the
generated type spec for that option will always be term/0, but you can use
this option to customize that. The value for this option must be a quoted Elixir
term. For example, if you have an :exception option that is validated with a
{:custom, ...} type (based on is_exception/1), you can override the type
spec for that option to be quote(do: Exception.t()). Available since v1.1.0.



  
    
  
  Types


	:any - Any type.

	:keyword_list - A keyword list.

	:non_empty_keyword_list - A non-empty keyword list.

	{:keyword_list, schema} - A keyword list matching the given options schema.

	:non_empty_keyword_list - A non-empty keyword list.

	{:non_empty_keyword_list, schema} - A non-empty keyword list matching the given options schema.

	:map - A map consisting of :atom keys. Shorthand for {:map, :atom, :any}.
Keys can be specified using the keys option.

	{:map, key_type, value_type} - A map consisting of key_type keys and
value_type values.

	:atom - An atom.

	:string - A string.

	:boolean - A boolean.

	:integer - An integer.

	:non_neg_integer - A non-negative integer.

	:pos_integer - A positive integer.

	:float - A float.

	:timeout - A non-negative integer or the atom :infinity.

	:pid - A PID (process identifier).

	:reference - A reference (see reference/0).

	nil - The value nil itself. Available since v1.0.0.

	:mfa - A named function in the format {module, function, arity} where
arity is a list of arguments. For example, {MyModule, :my_fun, [arg1, arg2]}.

	:mod_arg - A module along with arguments, such as {MyModule, arguments}.
Usually used for process initialization using start_link and similar. The
second element of the tuple can be any term.

	:fun - Any function.

	{:fun, arity} - Any function with the specified arity.

	{:fun, args_types} - A function with the specified arguments.

	{:fun, args_types, return_type} - A function with the specified arguments and return type.

	{:in, choices} or {:one_of, choices} - A value that is a member of one of the choices. choices
should be a list of terms or a Range. The value is an element in said
list of terms, that is, value in choices is true.

	{:struct, struct_name} - An instance of the struct type given.

	:struct - An instance of any struct

	{:tagged_tuple, tag, inner_type} - maps to {tag, type}

	{:spark_behaviour, behaviour} - expects a module that implements the given behaviour, and can be specified with options, i.e mod or {mod, [opt: :val]}

	{:spark_behaviour, behaviour, builtin_module} - Same as the above, but also accepts a builtin_module. The builtin_module is used to provide additional options for the elixir_sense plugin.

	{:spark_function_behaviour, behaviour, {function_mod, arity}} - expects a module that implements the given behaviour, and can be specified with options, i.e mod or {mod, [opt: :val]}, that also has a special module that supports being provided an anonymous function or MFA as the :fun option.

	{:spark_function_behaviour, behaviour, builtin_module, {function_mod, arity}} - Same as the above, but also accepts a builtin_module. The builtin_module is used to provide additional options for the elixir_sense plugin.

	{:behaviour, behaviour} - expects a module that implements a given behaviour.

	{:protocol, protocol} - expects a value for which the protocol is implemented.

	{:impl, protocol} - expects a module for which the protocol is implemented.

	{:spark, dsl_module} - expects a module that is a Spark.Dsl

	{:mfa_or_fun, arity} - expects a function or MFA of a corresponding arity.

	{:spark_type, module, builtin_function} - a behaviour that defines builtin_function/0 that returns a list of atoms that map to built in variations of that thing.

	{:spark_type, module, builtin_function, templates} - same as the above, but includes additional templates for elixir_sense autocomplete

	:literal -> any literal value. Maps to :any, but is used for documentation.

	{:literal, value} -> exactly the value specified.

	:quoted -> retains the quoted value of the code provided to the option

	{:wrap_list, type} -> Allows a single value or a list of values.

	{:custom, mod, fun, args} - A custom type. The related value must be validated
by mod.fun(values, ...args). The function should return {:ok, value} or
{:error, message}.

	{:or, subtypes} - A value that matches one of the given subtypes. The value is
matched against the subtypes in the order specified in the list of subtypes. If
one of the subtypes matches and updates (casts) the given value, the updated
value is used. For example: {:or, [:string, :boolean, {:fun, 2}]}. If one of the
subtypes is a keyword list or map, you won't be able to pass :keys directly. For this reason,
:keyword_list, :non_empty_keyword_list, and :map are special cased and can
be used as subtypes with {:keyword_list, keys}, {:non_empty_keyword_list, keys} or {:map, keys}.
For example, a type such as {:or, [:boolean, keyword_list: [enabled: [type: :boolean]]]}
would match either a boolean or a keyword list with the :enabled boolean option in it.

	{:list, subtype} - A list where all elements match subtype. subtype can be any
of the accepted types listed here. Empty lists are allowed. The resulting validated list
contains the validated (and possibly updated) elements, each as returned after validation
through subtype. For example, if subtype is a custom validator function that returns
an updated value, then that updated value is used in the resulting list. Validation
fails at the first element that is invalid according to subtype. If subtype is
a keyword list or map, you won't be able to pass :keys directly. For this reason,
:keyword_list, :non_empty_keyword_list, and :map are special cased and can
be used as the subtype by using {:keyword_list, keys}, {:non_empty_keyword_list, keys}
or {:keyword_list, keys}. For example, a type such as
{:list, {:keyword_list, enabled: [type: :boolean]}} would a list of keyword lists,
where each keyword list in the list could have the :enabled boolean option in it.

	{:tuple, list_of_subtypes} - A tuple as described by tuple_of_subtypes.
list_of_subtypes must be a list with the same length as the expected tuple.
Each of the list's elements must be a subtype that should match the given element in that
same position. For example, to describe 3-element tuples with an atom, a string, and
a list of integers you would use the type {:tuple, [:atom, :string, {:list, :integer}]}.
Available since v0.4.1.



  
    
  
  Example


iex> schema = [
...>   producer: [
...>     type: :non_empty_keyword_list,
...>     required: true,
...>     keys: [
...>       module: [required: true, type: :mod_arg],
...>       concurrency: [
...>         type: :pos_integer,
...>       ]
...>     ]
...>   ]
...> ]
...>
...> config = [
...>   producer: [
...>     concurrency: 1,
...>   ]
...> ]
...>
...> {:error, %Spark.Options.ValidationError{} = error} = Spark.Options.validate(config, schema)
...> Exception.message(error)
"required :module option not found, received options: [:concurrency] (in options [:producer])"

  
    
  
  Nested Option Items


Spark.Options allows option items to be nested so you can recursively validate
any item down the options tree.

  
    
  
  Example


iex> schema = [
...>   producer: [
...>     required: true,
...>     type: :non_empty_keyword_list,
...>     keys: [
...>       rate_limiting: [
...>         type: :non_empty_keyword_list,
...>         keys: [
...>           interval: [required: true, type: :pos_integer]
...>         ]
...>       ]
...>     ]
...>   ]
...> ]
...>
...> config = [
...>   producer: [
...>     rate_limiting: [
...>       interval: :oops!
...>     ]
...>   ]
...> ]
...>
...> {:error, %Spark.Options.ValidationError{} = error} = Spark.Options.validate(config, schema)
...> Exception.message(error)
"invalid value for :interval option: expected positive integer, got: :oops! (in options [:producer, :rate_limiting])"

  
    
  
  Validating Schemas


Each time validate/2 is called, the given schema itself will be validated before validating
the options.
In most applications the schema will never change but validating options will be done
repeatedly.
To avoid the extra cost of validating the schema, it is possible to validate the schema once,
and then use that valid schema directly. This is done by using the new!/1 function first, and
then passing the returned schema to validate/2.
Create the Schema at Compile Time
If your option schema doesn't include any runtime-only terms in it (such as anonymous
functions), you can call new!/1 to validate the schema and returned a compiled schema
at compile time. This is an efficient way to avoid doing any unnecessary work at
runtime. See the example below for more information.

  
    
  
  Example


iex> raw_schema = [
...>   hostname: [
...>     required: true,
...>     type: :string
...>   ]
...> ]
...>
...> schema = Spark.Options.new!(raw_schema)
...> Spark.Options.validate([hostname: "elixir-lang.org"], schema)
{:ok, hostname: "elixir-lang.org"}
Calling new!/1 from a function that receives options will still validate the schema each time
that function is called. Declaring the schema as a module attribute is supported:
@options_schema Spark.Options.new!([...])
This schema will be validated at compile time. Calling docs/1 on that schema is also
supported.

Spark.Options.Helpers 
    



      
Helpers for use with spark options

Spark.Options.ValidationError exception
    



      
An error that is returned (or raised) when options are invalid.
Since this is an exception, you can either raise it directly with raise/1
or turn it into a message string with Exception.message/1.
See %Spark.Options.ValidationError{} for documentation on the fields.

Spark.Options.Validator 
    



      
Defines a validator module for an option schema.
Validators create structs with keys for each option in their schema,
and an optimized validate, and validate! function on that struct.

  
    
  
  Upgrading from options lists


You can pass the option define_deprecated_access?: true to use Spark.Options.Validator,
which will make it such that options[:foo] will still work, but will emit a deprecation warning.
This cane help with smoother upgrades.

  
    
  
  Example


Given a module like the following:
defmodule MyOptions do
  use Spark.Options.Validator, schema: [
    foo: [
      type: :string,
      required: true
    ],
    bar: [
      type: :string
    ],
    baz: [
      type: :integer,
      default: 10
    ]
  ]
end
You can use it like so:
@doc """
Does a thing

## Options

#{MyOptions.docs()}
"""
@doc spark_opts: [{1, MyOptions.schema()}]
def your_function(arg, opts \\ []) do
  options = MyOptions.validate!(opts)

  options.foo
  options.bar
end

Spark.OptionsHelpers 
    



      
Helpers for working with options lists.

Spark.Error.DslError exception
    



      
Used when a 